



BBC Micro Listing

```

1000 REM **** BBC ROBOT MEASURE ****
1010 MODE 7
1020 PROCinitialise
1030 PROCmeasure
1040 PROCprintout
1050 END
1060 DEF PROCmeasure
1070 PROCfind
1080 REM ** ONE BUMPER ONLY ? **
1090 PROCTest_bumpers
1100 REM **** FIND END ****
1110 REPEAT:PROCprobe(right,width)
1120 UNTIL (?DATREG AND 192)=left_bumper
1130 REM ** GO BACK AND INCH TO END **
1140 PROCprobe(left,width)
1150 REPEAT:PROCprobe(right,small_width)
1160 UNTIL (?DATREG AND 192)=left_bumper
1170 PRINT"FOUND RIGHT-HAND END"
1180 PRINT"STARTING TO MEASURE"
1190 REM ** START TO MEASURE **
1200 count=width
1210 REPEAT:PROCprobe(left,width)
1220 count=count+width
1230 UNTIL (?DATREG AND 192)=right_bumper
1240 REM ** GO BACK AND INCH TO END **
1250 count=count-width
1260 PROCprobe(right,width)
1270 REPEAT:PROCprobe(left,small_width)
1280 count=count+small_width
1290 UNTIL (?DATREG AND 192)=right_bumper
1300 ?DATREG=0
1310 ENDPROC
1320 :
1330 DEF PROCprintout
1340 CLS
1350 PRINTTAB(5,12)"OBJECT SIDE MEASURED AT
";count;" mm"
1360 ENDPROC
1370 :
1380 DEF PROCinitialise
1390 DDR=&FE62:DATREG=&FE60
1400 ?DDR=15:REM LINES 0-3 OUTPUT
1410 ?DATREG=1:REM TURN ON RESET BIT
1420 forwards=4:backwards=2:left=6:right=0
1430 pd_ratio=3.34446:pa_ratio=375/90
1440 right_bumper=128:left_bumper=64
1450 both_bumpers=0:neither_bumpers=192
1460 width=60:small_width=5
1470 ENDPROC
1480 :
1490 DEF PROCsearch(sense)
1500 REPEAT:PROCprobe(sense,width)
1510 UNTIL (?DATREG AND 192)=both_bumpers
1520 ENDPROC
1530 :
1540 DEF PROCfind
1550 REPEAT:PROCmove(forwards,8)
1560 UNTIL(?DATREG AND 192)<neither_bumpers
1570 ENDPROC
1580 :
1590 DEF PROCTest_bumpers
1600 IF (?DATREG AND 192)=right_bumper THEN
PROCsearch(right):ENDPROC
1610 IF (?DATREG AND 192)=left_bumper THEN
PROCsearch(left):ENDPROC
1620 ENDPROC
1630 :
1640 DEF PROCprobe(way,step)
1650 IF way=right THEN opp_way=left ELSE
opp_way=right
1660 PROCmove(backwards,30)
1670 PROCturn(way,90)
1680 PROCmove(forwards,step)
1690 PROCturn(opp_way,90)
1700 REPEAT:PROCmove(forwards,8)
1710 UNTIL (?DATREG AND 192)<neither_bumpers
1720 ENDPROC
1730 :
1740 DEF PROCmove(dir,distance)
1750 ?DATREG=(?DATREG AND 1)OR dir
1760 pulses=pd_ratio*distance
1770 FOR I=1 TO pulses:PROCpulse:NEXT I
1780 ENDPROC
1790 :
1800 DEF PROCturn(dir,angle)
1810 ?DATREG=(?DATREG AND 1)OR dir
1820 pulses=pa_ratio*angle
1830 FOR I=1 TO pulses:PROCpulse:NEXT I
1840 ENDPROC
1850 DEF PROCpulse
1860 ?DATREG=(?DATREG OR 8)
1870 ?DATREG=(?DATREG AND 247)
1880 ENDPROC

```

Commodore 64 Listing

```

10 REM **** CBM ROBOT MEASURE ****
20 GOSUB1000:REM INITIALISE
30 GOSUB2000:REM MEASURE
40 GOSUB3000:REM PRINTOUT
50 END
60 :
1000 REM **** INITIALISE ****
1010 DDR=56579:DATREG=56577
1020 POKE DDR,15:REM LINES 0-3 OUTPUT
1030 POKE DATREG,1:REM TURN ON RESET BIT
1040 FW=4:BW=2:LF=6:RT=0
1050 PD=3.34446:PA=375/90
1060 RB=128:LB=64:BB=0:NB=192
1070 WD=60:SW=5
1080 RETURN
1090 :
2000 REM **** MEASURE ****
2010 GOSUB3500:REM FIND OBJECT
2020 GOSUB4000:REM TEST BUMPERS
2030 REM ** FIND END **
2040 WY=RT:SP=WD:GOSUB6000:REM PROBE
2050 IF(PEEK(DATREG)AND192)<<LB THEN 2040
2060 REM ** GO BACK AND INCH TO END **
2070 DR=LF:DS=WD:GOSUB6000:REM PROBE
2080 DR=RT:DS=SW:GOSUB6000:REM PROBE
2090 IF(PEEK(DATREG)AND192)<<LB THEN 2090
2100 PRINT"FOUND RIGHT-HAND END"
2110 PRINT"STARTING TO MEASURE"
2120 REM ** START TO MEASURE **
2130 CC=WD
2140 DR=LF:DS=WD:GOSUB6000:CC=CC+WD
2150 IF(PEEK(DATREG)AND192)<<RB THEN 2140
2160 REM ** GO BACK AND INCH TO END **
2170 CC=CC-WD
2180 DR=RT:DS=WD:GOSUB6000:CC=CC+SW
2190 IF(PEEK(DATREG)AND192)<<RB THEN 2180
2200 POKE DATREG,0
2210 RETURN
2220 :
3000 REM **** PRINT OUT ****
3010 PRINTCHR$(147)
3020 PRINT"OBJECT MEASURED AT ";CC;"MM"
3030 RETURN
3040 :
3500 REM **** FIND ****
3510 DR=FW:DS=5:GOSUB7000:REM MOVE
3520 IF(PEEK(DATREG)AND192)=NB THEN 3510
3530 RETURN
3540 :
4000 REM **** TEST BUMPERS ****
4010 IF(PEEK(DATREG)AND192)=RB THEN SS=RT:
GOSUB5000:RETURN
4020 IF(PEEK(DATREG)AND192)=LB THEN SS=LF:
GOSUB5000:RETURN
4030 RETURN
4040 :
5000 REM **** SEARCH (SS) ****
5010 DR=FW:DS=8:GOSUB7000:REM MOVE
5020 IF(PEEK(DATREG)AND192)=NB THEN 5010
5030 RETURN
5040 :
6000 REM **** PROBE (WY,SP) ****
6010 IF WY=RT THEN OW=LF
6020 IF WY=LF THEN OW=RT
6030 DR=BW:DS=30:GOSUB7000:REM MOVE
6040 DR=WY:AG=30:GOSUB7500:REM TURN
6050 DR=FW:DS=SP:GOSUB7000:REM MOVE
6060 DR=OW:AG=90:GOSUB7500:REM TURN
6070 DR=FW:DS=8:GOSUB7000:REM MOVE
6080 IF(PEEK(DATREG)AND192)=NB THEN 6070
6090 RETURN
6100 :
7000 REM **** MOVE (DR,DS) ****
7010 POKE DATREG,(PEEK(DATREG)AND 1)OR DR
7020 PL=PD+DS
7030 FOR I=1 TO PL:GOSUB8000:NEXT I
7040 RETURN
7050 :
7500 REM **** TURN (DR,AG) ****
7510 POKE DATREG,(PEEK(DATREG)AND 1)OR DR
7520 PL=PA+AG
7530 FOR I=1 TO PL:GOSUB8000:NEXT I
7540 RETURN
7550 :
8000 REM **** PULSE ****
8010 POKE DATREG,PEEK(DATREG)OR 8
8020 POKE DATREG,PEEK(DATREG)AND 247
8030 RETURN

```